::请使用4/29号之后的hdc_std工具
::设置编码方式
chcp 65001

@echo off
cd /d %~dp0

:: 创建目录结构
if not exist host (
	md host\fuzztest
	md host\unittest
	md host\tools
)

if not exist device (
	md device\fuzztest
	md device\unittest
	md device\tools
	echo 已完成目录结构创建，请准备好测试用例后，在执行我吧！
	pause
	exit
)

::设置不要息屏
hdc_std shell power-shell setmode 602

echo .
echo .
echo 使用前请确保目录结构如下:
echo ├─ device
echo │  ├─fuzztest
echo │  └─unittest	
echo ├─ host
echo │   ├─fuzztest
echo │   └─unittest
echo ├─ send_file.bat
echo └─ execute_test.sh
echo .
echo .
echo 1、发送Host端用例及工具
echo 2、发送device端用例及工具

set /p choiceRole= 请选择[1-host, 2-device]：

if "%choiceRole%"=="1" (
    set localDir=host
) else if "%choiceRole%"=="2" (
	set localDir=device
) else (
	echo 输入无效
	pause
	exit
)

set remoteRoot=/data/usbtest/


::创建板子上用例存放的目录
hdc_std shell mkdir -p %remoteRoot%

echo 正在发送目录%localDir%
::将测试文件推倒板子上的remoteDir目录下
hdc_std file send %localDir% %remoteRoot%

set execTestName=execute_test.sh

echo 正在发送文件%execTestName%
::将测试脚本推倒板子上
hdc_std file send %execTestName% %remoteRoot%
::给测试脚本赋予可执行权限
hdc_std shell chmod +x %remoteRoot%%execTestName%
echo .
echo .
echo 已将%localDir%目录推送到%remoteRoot%

pause
exit